
三角形の分割

和田 英一 （IIJ 技術研究所）
wada@u-tokyo.ac.jp

■区画内の点の個数

　図 -1に示すように二等辺直角三角形 ABCの内側に点が n個散在している（nは 3の倍数）．どの点も
各頂点からその点に引いた線分やその延長線上に他の点が載らないようにしてある．図で座標が整数値な
のは他意はない．今の条件を調べるのを楽にするためである．
　その三角形の内部にもう 1つの点 Qをたとえば (3, 3) にとり，図 -2のように各頂点からQへ線を引い
て三角形を 3つの区画に分割する．点が区画の境界にあれば，それにかかわる区画のいずれにも含まれ
るとする．そしてどの区画にも n3以上個の点が含まれるようにしたい．その Qの座標を答えるのが今
回の問題である．1998年，東京大会（@早稲田大学）の問題 F（問題文は http://www.isse.kuis.kyoto-u. 
ac.jp/acm-japan/contest.html参照）．

　図 -2から察せられるように，Qの位置には任意性があり，その Qの位置が各区画に本当に n3以上の
点が含まれるように分割するか，審判団は計算してチェックしたらしい．図 -3の 3点 P0, P1, P2のように
それぞれ各辺の中点のわずかに内側にあれば，人間到る処青山あり（どこにでも骨を埋められるの意）の
気分で，ほとんどどこもが Qの候補地である．正確にいえば各頂点からその角を挟む辺の中点付近の点
に引いた線で囲まれた 6角形 (P0, Q2, P1, Q0, P2, Q1)の中が解の存在領域である．
　区画内の点の個数だが，3点 P0(x0,y0), P1(x1,y1), P2(x2,y2)が構成する△ P0P1P2の面積 Sは行列式を使
うと

である．ただし P0→P1→P2が反時計回りか時計回りかでこの値は正か負になる（忘れたら (0, 1), (0, 0), 
(1, 0)の程度の三角形で検算する）．これを利用するが正負の情報だけ必要ゆえ，先頭の 12は今は使
わない．3点が 1直線上にあれば行列式の値は 0になる．分割点を Qとし，内部の点 Piに対して，△
AQPi, △ BQPi, △ CQPiの面積を計算し，その符号をこの順に左から書くと図 -4になる．これから分かる
ように△ BCQの区画内にある点 Pは B→Q→Pが時計回りなので負，C→Q→Pが半時計回りなので正に
なっている．一番左の符号はAQに対して Pが左にあるか右にあるかで負か正になり，この区画では問
題にならない．これを ?-+と表す．△ CAQ内の点 Pは +?-, △ABQ内の点 Pは -+?となっていること
が分かる．したがって Qが与えられたら，今のような符号の組を作り，?-+, +?-, -+?の個数を調べれば
各区画内の点の個数が判明する．

　今回のプログラムは lisp方言の schemeで書きたい．点の座標の並びなど，C言語の配列で書いたり，
C言語でリスト処理を書いて使ったりするより分かりやすいからである．
　まず下請け関数．crosspは点 p0, p1を結ぶ直線と点 p2, p3を結ぶ直線の交点の座標を返す．座標は xと
yのドット対．どうせ傾いている直線なので，気楽に書いてある．sideは p0, p1, p2が反時計回りか時計回
りか 1直線上にあるか判定し，1, 1, 0のいずれかを返す．countは△ BCq0, △ CAq1, △ABq2内にある
点の個数をこの順にリストにして返す．リストの要素がすべて n3個以上あれば分割成功である．

(define (crossp p0 p1 p2 p3)

  (let* ((x0 (car p0)) (y0 (cdr p0)) 

         (x1 (car p1)) (y1 (cdr p1))

         (x2 (car p2)) (y2 (cdr p2))

         (x3 (car p3)) (y3 (cdr p3))

         (a0 (- y1 y0)) (b0 (- x0 x1))

         (c0 (- (* x0 y1) (* x1 y0)))

         (a1 (- y3 y2)) (b1 (- x2 x3))

         (c1 (- (* x2 y3) (* x3 y2)))

         (d (- (* a0 b1) (* a1 b0))))

    (cons (/ (- (* c0 b1) (* c1 b0)) d)

          (/ (- (* a0 c1) (* a1 c0)) d))))

(define (side p0 p1 p2)

 (let* ((x0 (car p0)) (y0 (cdr p0)) (x1 (car p1)) (y1 (cdr p1))

        (x2 (car p2)) (y2 (cdr p2))

        (det  (- (+ (* x0 y1) (* x1 y2) (* x2 y0))

                 (+ (* x2 y1) (* x0 y2) (* x1 y0)))))

   (cond ((> det 0) 1)

         ((< det 0) -1)

         (else 0))))

(define (count q0 q1 q2)

   (let* ((sl (map (lambda (p)

                     (list (side (list-ref corners 0) q0 p)

                           (side (list-ref corners 1) q1 p)

                           (side (list-ref corners 2) q2 p))) points)))

    (list

      (length (filter (lambda (x) (and (<= (list-ref x 1) 0) 

                                       (>= (list-ref x 2) 0)))

                      sl))      ;(? - +) 下の△内の個数
      (length (filter (lambda (x) (and (<= (list-ref x 2) 0)

                                       (>= (list-ref x 0) 0)))

                      sl))      ;(+ ? -) 中の△内の個数
      (length (filter (lambda (x) (and (<= (list-ref x 0) 0)

                                       (>= (list-ref x 1) 0)))

                      sl)))))   ;(- + ?) 左の△内の個数

cornersという変数が見えるが，点や頂点は
(define corners '((0 . 10) (0 . 0) (10 . 0)))

(define points '((5 . 4) (4 . 5) (5 . 3) (3 . 5) (5 . 2) (4 . 3) 

                 (3 . 4) (2 . 5) (5 . 1) (1 . 5) (4 . 1) (3 . 2)

                 (2 . 3) (1 . 4) (3 . 1) (1 . 3) (2 . 1) (1 . 2)))

のように与えることにする．

■山登り法を真似る

　各区画内の点の個数の算出法が分かれば，とりあえず，溺れた場合に藁をも掴むが如き方法は考えら
れる．

図-5

　まず適当に分割中心を仮りに決め，分割を試み，その結果によって分割中心を移動修正するという方策
を考える．図 -5で「多」は n3以上の点を含む区画，「少」は n3未満の点を含む区画を表す．「多」「少」
をそれぞれ 1, 0で表し，区画を下，中（右上のこと），左の順に並べた 2進数のような表現が下に記して
ある．
　かくして 111になれば完成となる．さもなければ，矢印が示す方向に分割中心を移動する．移動の量や
その変更が秘妙なノウハウであって，さしたる自信はない．山登り法などと同様に私はこれを「東洋医学」
ふうだと思っている．
　今書いてあるプログラムは最初の移動量が直角を挟む辺の長さの 110で，向きが変わるたびに 0.8倍

しているが，必ずしもうまくいくという保証もない .
　図 -6に中心が移動する一例を示す．こういう方法で分散した点を含む最小の円の中心を求めるような
場合は最後の精度までつめよらなければならないが，この問題は適当な範囲で 3分割に成功するから，そ
ういう意味では楽かもしれない．
　私のプログラムでは図 -6の不思議な出発点ではなく，すべての点の位置の平均の位置から開始した．
図 -1のような対称的なデータだと平均の位置が (3, 3)になるから一発で的中する．審判団の用意したデー
タに，辺の長さ 1000で点の位置が (1,1), (2,1), …,(300,1)といういやらしいのがあったが，600数十ステッ
プで決着した．

(define (triangle)

  (let ((qx (/ (apply + (map car points)) (length points)))     ;x,yの初期値
        (qy (/ (apply + (map cdr points)) (length points)))

        (th (/ (length points) 3))      ;閾値
        (dx (/ (cdar corners) 10))      ;移動量
        (dy (/ (cdar corners) 10))

        (mode0 0)

        (mode1 0))

  (define (advance)

    (define (modify mode)

      (cond 

      ((= mode 1) (set! qx (max (- qx dx) 0.0)))        ;modeの値は図 -5と同じ
      ((= mode 2) (set! qx (+ qx dx)) (set! qy (+ qy dy)))

      ((= mode 3) (set! qy (+ qy dy)))

      ((= mode 4) (set! qy (max (- qy dy) 0.0)))

      ((= mode 5) (set! qx (- qx dx)) (set! qy (- qy dy)))

      ((= mode 6) (set! qx (+ qx dx)))))

   (define (dist count th)      ;2進→ 10進変換
     (let ((dist (map  (lambda (x) (if (< x th) 0 1)) count)))

       (+ (* (car dist) 4) (* (cadr dist) 2) (caddr dist))))

                                                        ;advanceの本体ここから
    (set! mode1 (dist (count (cons qx qy)(cons qx qy)(cons qx qy)) th))

    (cond ((= mode1 7) (cons qx qy))    ;うまく分割できたら座標を返す
          ((< mode1 7)

           (cond ((not (= mode1 mode0)) (set! mode0 mode1)

                                (set! dx (* dx 0.8))    ;移動量を減らす
                                (set! dy (* dy 0.8))

                                (modify (- 7 mode0))))  ;逆に 1回動く
           (modify mode1)       ;新しい方向に動く
           (advance))))         ;末尾再帰でループする
  (advance)))

　最初の letで変数を初期化し，あとは advanceでループしている．どの方向に進んでいるかはmodeが
覚えている．

■全数検査でやると

　こういう東洋医学手法は信念に合わないというときは確実に答えに至るアルゴリズムで解きたくなるで
あろう．
　それには次の方法を考えつく．
1. すべての点が分割点になり得ないか確かめる．
2. すべての相異なる 2点に頂点から引いた線の延長上の交点（図 -3の Qのような点）が分割点になり得
ないか確かめる．

　1. は点の個数に対してオーダ nであり，2. はオーダ n2である．1. では分割点になる点を提供した区画
はちょうど n3個含むことになるが，あとの 2区画は 1個ずつ余徳がつく．これは区画内個数算出の手続
きが用意できれば，プログラムは簡単だ．
　2. には解決すべきことがまだある．まず 2点を p0, p1として，それぞれにどの頂点から線を引くかだ．
答えは（v0, v1 ∈ A,B,C, p0, p1 ∈ P0,P1,…, Pn-1として）v0から p0, v1から p1に線を引くとき，p1は
v0p0に対して v1側にあり，p0は v1p1に対して v0側にあるべし．
　Aから p0に線を引くとする．p1が Ap0に対し B側にあれば p0が Bp1に対しA側にあればよい．つま
り，△Ap0p1  0 かつ △ Bp1p0  0 このときはAp0, Bp1を引く．反対に，p1がAp0に対し C側にあれ
ば p0が Cp1に対しA側にあればよい．つまり，△Ap0p1  0 かつ △ Cp1p0  0 このときはAp0, Cp1を
引く．Bや Cから p0に線を引く場合も同様に考えればよい．面積を 6種類計算しているみたいだが，△
vp0p1  △ vp1p0が利用できることに注意しよう．
　次は，各区画内の点の個数を調べるには，交点の座標ではなく，頂点からの勾配だけを知ればよいから，
たとえば図 -7で，Ap0と Cp2の延長線の交点 Dへ BDを引いたとき，その勾配を，Dの座標を明示的に
計算せずに得る方法の開発である．
　図 -7をみて考えよう．E, Gは Hからそれぞれ AB, BCに下ろした垂線の足，F, Hはそれを逆に伸ばし
て斜辺 CAとの交点．
　この図では 3点 p0, p1, p2の座標は p0(a, 10b), p1(c, d), p2(10e, f)のつもり．したがって a, b, …, f（そ
れぞれ 0, また b  a, e  f）は図に示すような長さになる．
　この図で重要なのは，直角二等辺三角形ゆえ，DFDHということ．さらに ED : DFa : ba, GD : 
DHf : ef である．この a, bと e, fのスケール比は ef : ba, 勾配は (a, bスケールの a):(e, fスケールの
f)である．∴ ED : GDc : da(ef) : f(ba).
　次にAp0, Bp1が先に与えられて，Cp2の勾配 e : fが欲しい場合はどうか．
　cf(ba)ad(ef)だから ef(ba)cf/ad, (ef)/fe/f1(ba)c/ad,
　∴ e : f(ba)cad : ad.
　Bp1, Cp2が先なら a : bは ba(ee)ad/cf, (ba)/ab/a1(ef)d/cf,
　∴ a : bcf : (ef)dcf.
　いま , a2, b5, c2, d3, e6, f3でチェックすると

　c : d2(63) : 3(52)23 : 332 : 3,
　e : f(52) 223 : 23326 : 612 : 62 : 1,
　a : b23 : (63)3236 : 3362 : 5,
　これを使えば交点をいちいち計算しないで済むが，問題に答えるには最後にどうしても交点の計算がい
る．2点 (p0, p1)と 2点 (p2, p3)を通る直線の交点を計算する crossを使う．しかしこういう工夫をしても，
計算時間はかなりかかり，予定時間を超えると打ち切られる恐れがある．

(define (triangle)

  (let* ((a (list-ref corners 0)) (b (list-ref corners 1)) 

         (c (list-ref corners 2)) (pleng (length points)) (plen3 (/ pleng 3))

         (plen1 (- plen3 1)) (lat (cdar corners)))

  (define (test1 p)                     ;点 pがうまく 3分割するかテスト
    (count p p p))

  (define (choosecorners p0 p1)         ;p0, p1がどの頂点と対応するか
    (let ((s0 (side a p0 p1)) (s1 (side b p0 p1)) (s2 (side c p0 p1)))

      (cond ((and (< s0 0) (< s1 0)) (cons 0 (list p0 p1 '()))) ;cornerへの点
            ((and (> s0 0) (> s2 0)) (cons 1 (list p0 '() p1)))

            ((and (< s1 0) (< s2 0)) (cons 2 (list '() p0 p1)))

            ((and (> s1 0) (> s0 0)) (cons 0 (list p1 p0 '())))

            ((and (< s2 0) (< s0 0)) (cons 1 (list p1 '() p0)))

            ((and (> s2 0) (> s1 0)) (cons 2 (list '() p1 p0))))))

  (define (test2 p0 p1)                 ;2点の延長線の交点が 3分割するか
    (let* ((corner (choosecorners p0 p1)) (ci (car corner))(cc (cdr corner)))

      (cond ((= ci 0)

             (let* ((a (car (list-ref cc 0))) (b (- lat (cdr (list-ref cc 0))))

                    (c (car (list-ref cc 1))) (d (cdr (list-ref cc 1)))

                    (e (+ (* (- b a) c) (* a d))) (f (* a d)))

               (count (cons a (- lat b)) (cons c d) (cons (- lat e) f))))

            ((= ci 1)

             (let* ((a (car (list-ref cc 0))) (b (- lat (cdr (list-ref cc 0))))

                    (e (- lat (car (list-ref cc 2))))

                    (f (cdr (list-ref cc 2)))

                    (c (* a (- e f))) (d (* f (- b a))))

               (count (cons a (- lat b)) (cons c d) (cons (- lat e) f))))

            ((= ci 2)

             (let* ((c (car (list-ref cc 1))) (d (cdr (list-ref cc 1)))

                    (e (- lat (car (list-ref cc 2))))

                    (f (cdr (list-ref cc 2)))

                    (a (* c f)) (b (+ (* (- e f) d) (* c f))))

               (count (cons a (- lat b)) (cons c d) (cons (- lat e) f)))))))

  (define (dist count th)

    (and (>= (car count) th) (>= (cadr count) th) (>= (caddr count) th)))

    (call-with-current-continuation     ;ループから飛び出す準備
      (lambda (exit)

        (do ((i 0 (+ i 1))) ((= i pleng))

          (let ((p (list-ref points i)))

            (cond ((dist (test1 p) plen3)

                   (exit p)))))         ;pが分割した．飛び出す．
        (do ((i 0 (+ i 1))) ((= i (- pleng 1)))

          (let ((p0 (list-ref points i)))

            (do ((j (+ i 1) (+ j 1))) ((= j pleng))

              (let ((p1 (list-ref points j)))

                (cond ((dist (test2 p0 p1) plen3)       ;p0 p1でテスト
                       (let* ((corner (choosecorners p0 p1))

                              (ci (car corner)) (cc (cdr corner)))

                         (exit (cond ((= ci 0)          ;もう一度交点を計算
                                      (crossp a (car cc) b (cadr cc)))

                                     ((= ci 1)

                                      (crossp a (car cc) c (caddr cc)))

                                     ((= ci 2)

                                      (crossp b (cadr cc) c (caddr cc))

  ))))))))))))))

　test1は各点が分割中心になるかを調べ（1. の場合），test2は p0, p1とそれぞれの対応する頂点を結
ぶ直線の交点が分割点になるかを調べ（2. の場合）ている．ループの途中で脱出したいので，call-with-
current-continuationを使っているが，それに続くラムダ式の変数（ここでは exit）を呼ぶと脱出する．
test2では脱出してから交点を計算する．

■敵は幾万ありとても

　東洋医学ふうではなくなったが，300点もあるとオーダ n2はやはり苦しい．無駄な計算をできるだけ省
き，解に向かって進みたいと思って工夫したのがこの算法である．図 -8で説明する．
図中の各点にβ , γという値を配する．点を Pとしてβは ∠ CBP, γは∠ACPである．
　2つのスィーパ，Cを中心に右回転する Scと，Bを中心に右回転する Sbがある．最初 Scは CBに，
Sbは BAに一致している．Sbと BCのなす角を Sβ , Scと CAのなす角を Sγと呼ぶことにする．Sβの初
期値は 90度，Sγは 45度である．
　Scは右回転を始め，点を n/3個通過した時点で停止する（図 -8では Cから放射線状に出ている線の一
番下）．この点のγの値を Sγとして，また点の番号をγインデックスとして記録する．次に Sbを右回転
し，Sγの線より下にある最初の点を探す（図では P17に到達し，Bから放射線状に出ている線の一番左に
いることになる）．この点のβを Sβとして，点の番号をβインデックスとして記録する．
　この時点で Bとβインデックスの点，Cとγインデックスの点を結ぶ線の延長線上の交点は n/3個の点
を含む．
　次に Scをさらに右回転し，Sβより小さいβを持つ最初の点を探す（図では P12が見つかる．Cからの

放射状の下から 2番目）．この点のγを Sγとして，γインデックスとともに記録する．今度は Sbの右回
転を続け，Sγより小さいγを持つ最初の点を探す（図では P12が見つかる）．この点のβを Sβとし，Bイ
ンデックスとともに記録（このように 2つの点の延長線上の交点ではなく，1つの点になることもある）．

　この作業を Scが CAに到達するまで繰り返す．図でいえば 4本の Sbと 4本の Scが得られる．Sbの
右から i番目と Scの下から i番目が対になり，この各対で囲まれた区画は n/3個の点が含まれる．この
各対の交点で三角形を 3つに分割した場合，各区画に n/3個以上の点が含まれれば，分割に成功したこ
とになる．この方法では Scは点のγ値の順にソートしたリストを，γ値の大きい方から小さい方へ順に 1
回辿るだけで作業が終るので，これは一見オーダ nの算法に見えるが，中でソートを使っているので，そ
れよりは悪い．
　しかし待てよ，と思わぬでもない．もう一度図 -8を見よう．n/3個の点を含む 2つのスィーパの組の相
隣る交点，たとえば P12（Q0としよう），それと P11と P4の延長線上の交点（Q1としよう）に着目する．
それらの点とスィーパの軸でなかった頂点（今はA）を結ぶ図の 2本の破線の間の領域に多くの点が散在
し，Q0を選ぶとAQ0の右には多数の点が含まれ，次にAQ1を選ぶと，今度はその分割線の左に多くの
点が含まれ，ちょうど半々にならないのではないかという疑問は当然出てくる．
　図 -9の点は (2, 1), (5, 0.5), (8, 1), (2, 5), (2.5, 5), (3, 5)に取ってある．辺 BCの上に 2点を含む三角形を
作ると △ BP0C, △ BP2Cができる（2本の線が重なることを破線で示す）．しかし線分 AP0は左が 1個，
右が 6個と，線分 AP2は左に 6個，右に 0個と分割するので，どちらでもだめである．そうなると次は
頂点 CとAでスィープするか，頂点Aと Bでスィープするかしなければならない．
　以下のプログラムはパラメータ vを 0, 1, 2と変えて 3つの方向でテストするようにできている．しかし
本体はβとγを使うように書いてあるから，上の説明とあっている．実際には vを変えるとα， β， γが
循環して使われる．

(define qpi (atan 1))           ;π /4

(define (triangle)

  (let* ((pleng (length points)) (plen3 (/ pleng 3)) (plen1 (- plen3 1))

         (lat (cdar corners)) (cc '()) (pointlist '()) (bdeg '()) (gdeg '())

         (blist '()) (gindex '()) (bindex '()) (gammalist '()))

  (define (v-iter v)            ;頂点の対を順にテストする . v=0なら B, C. 

    (let ((v1 (+ (remainder (+ v 1) 3) 3)) (v2 (+ (remainder (+ v 2) 3) 3))

        (bs (list (* 2 qpi) qpi qpi (* 2 qpi)))) 

  

    (define (gfind gg)

      (if (null? gg) '()

          (let* ((p (list-ref pointlist (caar gg)))

                 (g (list-ref p v2)) (b (list-ref p v1)) (i (list-ref p 2)))

            (cond ((and (< b bdeg) (< g gdeg))  ;点を取り込む条件
                   (set! gindex (list-ref p 2)) (set! gdeg (list-ref p v2))

                   (set! blist 

                         (cons (list (list-ref p 2) (list-ref p v1)) blist))

                   '#t)

                  (else (gfind (cdr gg)))))))   ;末尾再帰
  

    (define (update)

      (set! blist (sort blist (lambda (x y) (> (cadr x) (cadr y)))))

      (set! bdeg (cadar blist))

      (set! bindex (caar blist))

      (set! blist (cdr blist))

      (set! cc (crossp (list-ref corners (- v1 3)) (list-ref points bindex) 

                       (list-ref corners (- v2 3)) (list-ref points gindex))))

  

    (define (g-iter)            ;γ角を順に調べるための反復
      (cond ((= (length (filter (lambda (x) (>= x plen3)) (count cc cc cc))) 3)

             cc)                ;うまく 3分割できたら交点の座標をもって帰る
            (else (cond ((gfind gammalist) (update) (g-iter))

                        (else '())))))

  

    (set! gammalist     ;v-iterの本体 ここから γ角の順にソートしたリストを作る
      (sort

        (map (lambda (x) (list (list-ref x 2) (list-ref x v2))) pointlist)

        (lambda (x y) (> (cadr x) (cadr y)))))

    (set!  bdeg (list-ref bs v))        ;β角の初期化
    (set!  gdeg (list-ref bs (+ v 1)))  ;γ角の初期化
    (set!  blist '())                   ;βリストの初期化
    (do ((j 0 (+ j 1))) ((= j plen3))

      (gfind gammalist))                ;まず n/3個の点を確保
    (update)

    (cond ((g-iter))

          (else (v-iter (+ v 1))))))    ;v-iterの末尾再帰
 

  (define  i -1)        ;triangleの本体 ここから
  (set!  pointlist

     (map (lambda (p)                   ;各点のα , β , γを計算する．
            (let* ((x (car p)) (y (cdr p))

                   (alpha (atan (/ x (- lat y))))

                   (beta (atan (/ y x)))

                   (gamma (- (atan (/ (- lat x) y)) qpi)))

               (set! i (+ i 1)) (list x y i alpha beta gamma)))

           points))

  (v-iter 0)))          ;まず辺 BCに対して v-iterを呼ぶ

点が 300個ある審判団のデータでも，すぐに答えが得られて，快適である．

■所要時間

　それぞれのプログラムで実験した所要時間（秒）を表 -1に示す．

点の個数
図 -1のデータ

18
審判団データ 1

300
審判団データ 2

300
山登り法 0 32 1.5
全数検査 0.25 1200 335
敵は幾万 0 7.3 3.1

表 -1

■なぜうまく分割できるか

　もとの問題文には，こういうふうに分割できることは証明可能としか書いてない．しかしなぜ分割可能
か少し考えてみたい．図 -10は図 -8をすべての頂点の組に対して書いたものである．2本の線が重なると
ころを黒の破線で示すのは図 -9と同じである．

　2つの頂点と内部の他の 1点，または 2点で構成する三角形は，ちょうと n/3個の点を含む最小の三角
形である．たとえば B P17 P11 Cには P6, P10, P11, P14, P16, P17の 6個の点がある．B P12 Cでは P17を失っ
た代わりに P12を得た．B P11 P4 Cでは P12を失った代わりに P4を得た．B P2 Cでは P11を失った代わ
りに P2を得た．つまり常に 6個の点がある．B, Cを頂点としちょうど 6個の点がある三角形はこれだけ
だ．他の頂点の組についても同様．
　さて全部で点が n個あって，各辺に立つ小三角形にそれぞれ n/3個の点があれば，各辺から適切に小
三角形を選べば，各点がいずれかの小三角形に含まれるように選べる．いまの例では (B P11 P4 C), (C P5 
P6 A), (A P7 P12 B)を選べばよい．そうしたら各頂点から，選ばれた小三角形の辺の中間へ向かって分割
線を引き，その 3本の交点を Qとすればよい．
　もう一度図 -9を見てみよう．この場合は最小三角形の取り方は一意ではなく，
(B P0 C), (C P2 P5 A), (A P4 B)

の組と
(B P2 C), (C P4 A), (A P3 P0 B)

の組が存在する . 

■問題の難易度はどうか

　この問題はやってみると見かけに相違して，存外手ごわかったという感じである．プロムナードの原稿
としては，こればかりにとりかかっているわけではないが，たっぷり 1カ月くらい，あれこれ考えさせら
れた． 
　すべてがこれほど手ごわいとも思わないが（この問題は Fであった），本番のコンテストでは 5時間に
8問題解くのだからコンテストに参加する学生さんは大変である．プログラムは審判団の用意したテスト
データに対して規定の時間内にしかるべき答えが出力できればよく，多少違っていてもそのテストにさえ
合格すればしめたものというわけだが，特に今回のような問題は，本来ならプログラムを書くときの方針
とか，プログラムの読みやすさ，きれいさなども評価するのが望ましい．しかしさて実行のことを考える
と不可能であろう．
　今回はどこかにも書いたように schemeのプログラムを示すことで勘弁してもらった．私はだいたいい
つも utilispでプログラムを書く，あるいは考える．しかし今回は幾何学的な問題なので，プログラムは
postscriptでかなり書いた．最後のスィープのプログラムもオリジナルは postscriptである．プログラム
を書きながら，少しずつ描画して確認作業をした．したがって postscript, utilisp, schemeと 3種類のプロ
グラムが手元にそろっている．そのうち schemeのはなるべく読みやすくするように努めて書き直しを繰
り返したが，まだ納得のいくレベルにはならなかったのは気になっている．
　8月に亡くなった Dijkstraならどういうプログラムを書いただろうか．見てみたいものである．

（平成 14年 10月 24日受付）


